SAS® GLOBAL FORUM 2020 #SASGF

Paper 4156-2020

Using the Optgraph Procedure to Construct Closed Knight's Tours on Standard and Variant Chessboards

Joe DeMaio, Kennesaw State University;
Md Shafiul Alam, Kennesaw State University;

ABSTRACT

A closed knight's tour of a chessboard uses legal moves of the knight to visit every square exactly once and return to its starting position. The closed knight's tour is a recreational variation of the well-known Traveling Salesman Problem (TSP). Gerlach uses Warnsdorff's heuristic technique and SAS® to produce closed knight's tours for the traditional $\boldsymbol{8 \times 8}$ board. Generalizing the same approach, Gerlach and Gerlach produce closed knight's tours for the $\boldsymbol{n \times 8 \times 8}$ chessboard for $\boldsymbol{n = 2, 3, 4, 5, 6}$ and theorize that the technique works for larger values of *n* as well. In 1991, Schwenk completely classified the $\boldsymbol{n \times m}$ rectangular chessboards that admit a closed knight's tour. In 2011, DeMaio and Mathew completely classified the $\boldsymbol{i \times j \times k}$ rectangular prism chessboards that admit a closed knight's tour. This paper demonstrates how to utilize PROC OPTGRAPH with the TSP option to construct closed knight's for rectangles, rectangular prisms, hexagons and other variant chessboards.

INTRODUCTION

The traveling salesman problem (TSP) wishes to construct a route that starts a salesperson at a home base, visits every client on a list exactly once, returns to the starting location and does so as cheaply as possible. When posed for a geographical region, travel between any two locals is possible with a cost specific to the pair. Many different routes for a region exist; the question is which route is the cheapest. At other times, the TSP is a question of existence; does a route that visits each location exactly once even exist? Such is the case of the closed knight's tour. The knight cannot legally move from just any given square to any other square. Furthermore, there is no cost associated with a knight's move. Can a knight use legal moves to visit every square on the chessboard and return to its starting position? Naturally, the question was initially solved for the standard 8×8 board. In 2015, Gerlach implements Warnsdorff's heuristic technique in SAS to produce closed knight's tours for the traditional 8×8 board [4]. When attempting to iteratively build a closed knight's tour, Warnsdorff's heuristic selects the next move to be the square with the fewest neighbors left unvisited in the current tour construction [7].

Typically, graphs or networks model the TSP problem. This comes as no surprise as Euler is considered the father of graph theory [2] whose 1759 paper [3] applies graphs as a model to tackle the problem of the closed knight's tour. Formally, a graph $G = (V, E)$ is an ordered pair where the vertex set V represents a set of squares on the board and the edge set E represents a legal move of the knight between a pair of squares. A closed knight's tour of a board is a cycle in the graph that visits every vertex exactly once and returns to its starting position. We know a closed knight's tour exists on the 8×8 board. What about rectangular boards of different dimensions?

CLOSED KNIGHT'S TOURS ON THE RECTANGULAR CHESSBOARD

In 1991, Schwenk classified all rectangular boards that admit a closed knight's tour [6].

Schwenk's Theorem: An $m \times n$ chessboard with $m \leq n$ contains a closed knight's tour unless one or more of the following three conditions hold:

(a) m and n are both odd;

(b) $m \in \{1, 2, 4\}$;

(c) $m = 3$ and $n \in \{4, 6, 8\}$.

It can be easily shown why there is no closed knight's tour on an $m \times n$ chessboard, where m and n are both odd. On the black and white chessboard, a legal move of the knight is possible, only between squares of alternating colors. Clearly, we need to have an equal number of black and white squares for a closed knight's tour. If m and n are both odd, the number of vertices on the board is also odd. Thus, there cannot be the same number of black and white squares and a closed knight's tour is not possible. A similar, but more complex, argument shows that no tour exists of the $4 \times n$ chessboard.

Let's consider a different approach on the non-existence of a knight's tour on a small board. Specifically, consider the 3×6 chessboard in Figure 1.



Figure 1. The non-existence of a closed knight's tour on the 3×6 chessboard

First, examine the squares $(2,1)$ and $(2,5)$. The only two possible moves starting from or ending at $(2,1)$ are to or from $(1,3)$ and $(3,3)$. Second, note that $(2,5)$ is also forced to have moves to or from $(1,3)$ and $(3,3)$. No other moves exist at $(2,1)$ and $(2,5)$. We must have a closed tour including these four vertices. There is no way to open up this tour to include the other squares. So, a 3×6 chessboard cannot admit a closed knight's tour. Similar arguments preclude the existence of a closed knight's tour for the remaining pairs of values in Schwenk's Theorem.

Proving the existence of a closed knight's tour for all other pairs of values is a theoretical exercise using induction as a proof technique. What if you only want to find a closed knight's tour for a specific rectangular board that we know exists? There is no need to go into the theoretical construction from Schwenk's paper. PROC OPTGRAPH offers many algorithms for graph analysis. Key to easily constructing closed knight's tour is the TSP option.

USING PROC OPTGRAPH TO CONSTRUCT CLOSED KNIGHT'S TOURS ON THE RECTANGULAR CHESSBOARD

Constructing a closed knight's tour utilizing PROC OPTGRAPH requires two steps. First, define the vertices and edges of the graph based on the type of chessboard and legal moves of the knight. In 2019, DeMaio and Yockey provide the following code for the general $m \times n$ chessboard [9]. First, define the specific values for m and n in the rows and columns variables. Second create the vertices and legal knight moves. In terms of total number of squares, the smallest boards that admit a closed knight's tour are the 3×10 and 5×6 chessboards. The following code creates a tour for the 5×6 chessboard. The code is easily

changed for any chessboard by redefining the variables for `rows` and `columns`. The graph is constructed by creating edges based on legal moves of the knight where squares on the board are represented in standard matrix format as ordered pairs. An edge need only be constructed once. The move connecting (3,1) and (1,2) is only created once by the following code as edges are created from left to right.

```
data board;
rows = 5;
columns = 6;
do i = 1 to rows;
do j = 1 to columns;
origin = (i || j);
/* input legal moves of the knight */
/* up 2 right 1 */
if i>=3 and j<= columns-1 then do
destination = (i-2 || j+1);
output;
end;
/* up 1 right 2 */
if i>=2 and j<= columns-2 then do
destination = (i-1 || j+2);
output;
end;
/* down 1 right 2 */
if i<=rows-1 and j<= columns-2 then do
destination = (i+1 || j+2);
output;
end;
/* down 2 right 1 */
if i<=rows-2 and j<= columns-1 then do
destination = (i+2 || j+1);
output;
end;
end;
end;
run;
```

Now that the chessboard is represented as a graph, we utilize the TSP option in PROC OPTGRPAH to construct a closed knight's tour.

```
proc optgraph
data_links = board;
*create graph from variables;
data_links_var
from = origin
```

```
to = destination;
* write cycle to file closed_knights_tour;
tsp out = closed_knights_tour;
run;
proc print data=work.closed_knights_tour;
* print to results file;
run;
```

Running PROC OPTGRAPH created the following closed knight's tour on the 5×6 board in the Work directory as `closed_knights_tour` as shown in Table 1. Figure 2 provides a visual representation of said tour.

Performance Information	
Execution Mode	Single-Machine
Number of Threads	4

Problem Summary	
Input Type	Graph
Number of Nodes	30
Number of Links	62
Graph Direction	Undirected

Solution Summary	
Problem Type	Traveling Salesman Problem
Solution Status	Optimal
Objective Value	30
Relative Gap	0
Absolute Gap	0
Primal Infeasibility	0
Bound Infeasibility	0
Integer Infeasibility	0
Best Bound	30
Nodes	0
Iterations	0
CPU Time	0.00
Real Time	0.00

Obs	origin		destination	
1	1	1	2	3
2	4	2	2	3
3	2	1	4	2
4	2	1	1	3
5	1	3	2	5
6	2	5	4	6
7	5	4	4	6
8	3	3	5	4
9	1	2	3	3
10	3	1	1	2
11	3	1	5	2
12	5	2	4	4
13	4	4	5	6
14	3	5	5	6
15	3	5	1	6

Obs	origin		destination	
16	2	4	1	6
17	2	4	4	5
18	4	5	2	6
19	1	4	2	6
20	2	2	1	4
21	4	1	2	2
22	4	1	5	3
23	5	3	3	4
24	3	4	1	5
25	1	5	3	6
26	5	5	3	6
27	4	3	5	5
28	5	1	4	3
29	5	1	3	2
30	1	1	3	2

Table 1. Results of PROC OPTGRAPH on the 5×6 chessboard

1	10	5	20	25	16
4	21	2	17	6	19
11	30	9	24	15	26
22	3	28	13	18	7
29	12	23	8	27	14

Figure 2. A closed knight's tour on the 5×6 chessboard

When running the above code to find a closed knight's tour when we know one exists, we merely need to look at the output file in the work directory to find a tour. What if our `rows` and `columns` variables are for a board where no tour exists? Two different cases are possible, and both require looking at the results output.

The most common case occurs on a board where legal knight moves exist from every square on the board, but no overall tour exists. This would be the case for the 5×7 board. Obviously, no tour is generated in the work directory. Instead one needs to look at the Solution Summary of the TSP option. The absence of a tour for the 5×7 board is indicated by the Solution Status set to Infeasible as seen in Table 2.

Performance Information	
Execution Mode	Single-Machine
Number of Threads	4

Problem Summary	
Input Type	Graph
Number of Nodes	35
Number of Links	76
Graph Direction	Undirected

Solution Summary	
Problem Type	Traveling Salesman Problem
Solution Status	Infeasible
CPU Time	0.00
Real Time	0.02

Table 2. Results of PROC OPTGRAPH on the $5 \times$7 chessboard

The rare second case occurs on a board that contains a square which has no legal knight moves. This takes place on the 3×3 board for the (2,2) square. Since the board is created using legal moves of the knight, the (2,2) square is not represented by a vertex. For the 3×3 board, a closed tour of the eight other squares does exist and is constructed since the $(2,2)$ square is not part of the graph. This case is rare but easy to note as the number of squares on the board, 9, is not the same as the number of nodes in the graph which is 8 as shown in the Problem Summary in Table 3. This case only occurs with small parameters of a board where a square has no room to make a legal knight move.

Performance Information	
Execution Mode	Single-Machine
Number of Threads	4

Problem Summary	
Input Type	Graph
Number of Nodes	8
Number of Links	8
Graph Direction	Undirected

Solution Summary	
Problem Type	Traveling Salesman Problem
Solution Status	Optimal
Objective Value	8
Relative Gap	0
Absolute Gap	0
Primal Infeasibility	0
Bound Infeasibility	0
Integer Infeasibility	0
Best Bound	8
Nodes	0
Iterations	0
CPU Time	0.00
Real Time	0.00

Obs	origin		destination	
1	1	1	2	3
2	3	1	2	3
3	3	1	1	2
4	1	2	3	3
5	2	1	3	3
6	2	1	1	3
7	3	2	1	3
8	1	1	3	2

Table 3. Results of PROC OPTGRAPH on the 3×3 chessboard



Figure 3. A partial closed knight's tour on the 3×3 chessboard

USING PROC OPTGRAPH TO CONSTRUCT CLOSED KNIGHT'S TOURS ON THE RECTANGULAR PRISM CHESSBOARD

Having studied the existence of closed knight's tour for the rectangular chessboard, it makes sense to consider other variant chessboards. In 2018, using the same heuristic technique in SAS, Gerlach and Gerlach produced closed knight's tours for the $\boldsymbol{n \times 8 \times 8}$ chessboard for $\boldsymbol{n = 2, 3, 4, 5, 6}$ and theorize that the technique works for larger values of *n* as well [5]. Naturally the question of existence of closed knight's tours on such a chessboard arises. In 2011, DeMaio and Mathew classified all rectangular prisms that admit a closed knight's tour [1].

An $i \times j \times k$ chessboard for integers $i, j, k \geq 2$ has a closed knight's tour unless one or more of the following three conditions hold:

(a) *i*, *j* and *k* are all odd;

(b) $i = j = k = 2$;

(c) $i = 2$ and $j = k = 3$.

Running PROC OPTGRAPH created the closed knight's tour on the $2 \times 4 \times 3$ board of Table 4. The code is available in Appendix A.

Obs	origin	destination
1	1 1 1	2 1 3
2	1 3 3	2 1 3
3	1 1 2	1 3 3
4	1 1 2	2 3 2
5	2 1 1	2 3 2
6	1 3 1	2 1 1
7	1 3 1	2 3 3
8	1 1 3	2 3 3
9	1 1 3	1 2 1
10	1 2 1	2 4 1
11	1 4 3	2 4 1
12	1 4 3	2 2 3
13	2 2 3	2 4 2
14	1 2 2	2 4 2
15	1 2 2	1 4 1
16	1 4 1	2 2 1
17	1 2 3	2 2 1
18	1 2 3	1 4 2
19	1 4 2	2 2 2
20	2 2 2	2 4 3
21	2 3 1	2 4 3
22	2 1 2	2 3 1
23	1 3 2	2 1 2
24	1 1 1	1 3 2

Table 4. Results of PROC OPTGRAPH on the $2 \times 4 \times 3$ chessboard

We omit the detailed discussion of output for rectangular prisms as it is identical to that for rectangular boards. As before, there are boards that admit a closed knight's tour, boards that don't due to a different number of black and white squares, and boards that are too small to admit a tour or too small to allow a knight move from particular cells.

USING PROC OPTGRAPH TO CONSTRUCT CLOSED KNIGHT'S TOURS ON THE RECTANGULAR PRISM CHESSBOARD

The closed knight's tour always accompanies the numerous variations of chessboards that exist. One need only redefine the topology of the chessboard and the moves of the knight for PROC OPTGRAPH. Many chessboard variants utilize a hexagonal cell rather than a square. One of the more popular of these hexagonal variants is Glinski's [8].



Figure 4. Glinski's hexagonal board of side 6 and knight's move

As we begin an examination of the closed knight's tour problem, we focus on a single parameter in much the same way one studies square chessboards before moving on to rectangular chessboards. One quickly notes a lack of right angles. Glinski's knight moves on an obtuse angle.

Glinski's hex boards of side 1, 2, or 3 are too small to permit a closed knight's tour since there are no legal moves from the center square. On the board of side 4, the knight may move from the center square and perhaps a closed tour does exist. Running PROC OPTGRAPH created the tour on the board of Figure 5. The code is available in Appendix B.



Figure 5. A closed knight's tour on Glinski's board of side 4

Repeating this process we established that a closed knight's tour exists on Glinski's hexagonal board of side *n* for $4 \leq n \leq 50$. PROC OPTGRAPH has been invaluable in demonstrating the existence of closed knight's tours in Glinski's hex board. Thus, we propose the following.

Conjecture: Glinkski's hexagonal board of side *n* contains a closed knight's tour for $n \geq 4$.

To prove this conjecture one must venture into the theoretical process of mathematical induction. Mathematical induction details the process of using knight's tours from smaller boards to create knight's tours for larger boards. The general nature of the construction will show how to create a tour on and board of side *n*. Doing so is the subject of a future paper.

CONCLUSION

PROC OPTGRAPH reduces the problem of constructing a TSP route down to the problem of coding the network. On the recreational and theoretical side of life, this paper uses PROC OPTGRAPH to construct closed knight's tours on variant chessboards. Elegantly coding the topology of the chessboard and moves of the knight serves as the challenge here. On the applied side of life, data for the TSP will be messy and big but eventually will be cleaned and formatted as a spreadsheet. After that importing the data and constructing the graph for use with PROC OPTGRAPH will be easy.

REFERENCES

[1] DeMaio, J. and Mathew, B. 2010. "Which Chessboards have a Closed Knight's Tour within the Rectangular Prism?". The Electronic Journal of Combinatorics, Volume 18, Issue 1(2011).

[2] Euler, Leonhard. 1736. "Solutio problematis ad geometriam situs pertinentis". *Comment. Acad. Sci. U. Petrop* 8, 128–40.

[3] Euler, Leonhard. 1759. "Solution d'une question curieuse que ne paroit soumise à aucune analyse". Mémoires de l'académie des sciences de Berlin 15 (1759) 1766 p. 310-337

[4] Gerlach, John R. 2015. "The Knight's Tour in Chess – Implementing a Heuristic Solution". SAS Global Forum, 2015.

[5] Gerlach, John R. 2019. "The Knight's Tour in 3-Dimensional Chess". SAS Global Forum, 2019.

[6] Schwenk, A. J. 1991. "Which Rectangular Chessboards have a Knight's Tour?". Mathematics Magazine 64:5 (December 1991) 325-332.

[7] Warnsdorff, H. C. 1823. *Des Rösselsprungs einfachste und allgemeinste Lösung*. Schmalkalden.

[8] Wikipedia Contributors. Hexagonal chess. "Gliński's hexagonal chess". Accessed September 24, 2019. https://en.wikipedia.org/wiki/Hexagonal_chess.

[9] Yockey, B. and DeMaio, J. 2019. "Using Proc Optgraph to implement the Prize Collecting Traveling Salesman Problem in SAS (Gotta catch as many as we can in a Pokémon raid for Alice)". SAS Global Forum, 2019.

ACKNOWLEDGMENT

We would like to thank SAS Global Forum for inviting us to share our interest in closed knight's tours and the use of PROC OPTGRAPH in optimization problems. Thanks are also extended to Kennesaw State University's Department of Statistics and Analytical Sciences

for funding travel to SAS Global Forum. None of this would be possible without the Analytics and Data Science Institute, as championed by Dr. Jennifer Priestley, for bringing SAS to the Kennesaw State campus. At the machine level both Devin Pearson and Chris Dow provided amazing technical support for the SAS grid and PROC OPTGRAPH.

RECOMMENDED READING

- Gould, R. J. Recent advances on the Hamiltonian problem: Survey III, *Graphs and Combinatorics* (2014) 30:1-46
- Watkins, J. J., & Rzewnicki, A. *Across the board: The mathematics of chessboard problems*. Princeton, Princeton University Press (2012)

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the authors at:

Joe DeMaio
Kennesaw State University
jdemaio@kennesaw.edu

Md Shafiul Alam
Kennesaw State University
malam6@students.kennesaw.edu

APPENDIX A-RECTANGULAR PRISM BOARD CODE

The following code creates a tour for the $4 \times 4 \times 4$ chessboard. The code is easily changed for any chessboard by redefining the variables for `rows` and `columns` and `layers`. The graph is constructed by creating edges based on legal moves of the knight where squares on the board are represented in standard matrix format as ordered triples. An edge need only be constructed once. The move connecting (1,3,1) and (1,1,2) is only created once by the following code as edges are created from left to right and up.

```
data board;
rows = 4;
columns = 4;
layers = 4;
DO I = 1 TO rows;
do j= 1 to columns;
do k = 1 to layers;
origin = (i||j||k);

/*Legal moves*/
/*1: Up 2 Right 1 same layer*/
if i>=3 and j <= columns - 1 then do
destination = (i-2||j+1||k);
output;
end;

/*2: Up 1 Right 2 same layer*/
if i>=2 and j<=columns - 2 then do
```

```
destination = (i-1||j+2||k);
output;
end;

/*3: Down 1 Right 2 same layer*/
if i<=rows-1 and j<=columns - 2 then do
destination = (i+1||j+2||k);
output;
end;

/*4: Down 2 Right 1 same layer*/
if i<=rows - 2 and j<=columns - 1 then do
destination = (i+2||j+1||k);
output;
end;

/*5: Up 2 down 1 layer*/
if i>=3 and k <= layers - 1 then do
destination = (i-2||j||k+1);
output;
end;

/*6: Up 1 down 2 layers*/
if i>=2 and k<=layers - 2 then do
destination = (i-1||j||k+2);
output;
end;

/*7: Down 1 down 2 layers*/
if i<=rows-1 and k<=layers - 2 then do
destination = (i+1||j||k+2);
output;
end;

/*8: Down 2 down 1 layer*/
if i<=rows - 2 and k<=layers - 1 then do
destination = (i+2||j||k+1);
output;
end;

/*9: Right 1 down 2 layers*/
if j<=columns-1 and k<=layers - 2 then do
destination = (i||j+1||k+2);
output;
end;

/*10: Right 2 down 1 layer*/
if j<=columns - 2 and k<=layers - 1 then do
destination = (i||j+2||k+1);
```

```
output;
end;

/*11: Right 1 up 2 layers*/
if j<=columns-1 and k>=3 then do
destination = (i||j+1||k-2);
output;
end;

/*12: Right 2 up 1 layer*/
if j<=columns - 2 and k>=2 then do
destination = (i||j+2||k-1);
output;
end;

end;
end;
end;
run;

proc optgraph
data_links = board;
data_links_var
from =origin
to = destination;
tsp out = latest_closed_knights_tour;
run;
```

APPENDIX B-GLINSKI'S HEXAGONAL BOARD CODE

The following code creates a tour for Glinski's hexagonal board of side 4. The code is easily changed for any chessboard by redefining the variable for `dim`. The graph is constructed by creating edges based on legal moves of the knight where squares on the board are represented in standard matrix format as ordered pairs. Given the complex topology of the board and movement of the knight, all possible moves of the knight are created for each hex cell. Thus, duplicate edges will be created. PROC OPTGRAPH ignores multiple edges in a graph although it does create a warning of each instance in the log.

```
data board;
dim = 4;
n = (dim - 1);
/* number of rows when dimension is 4 = 7 that is (4-1)*2 + 1 */
/* 0 to (4-1)*2 iterates 7 times: r= 0,1,2,3,4,5,6  */
do r = 0 to 2*n;

    /*case when dimension = 4, n = 3, r = 0, n-r = 3, column starts at 3 ends at 6*/
    /*                                                                    r = 1, n-r = 2, column starts at
2 ends at 6*/
```

```
	/*                                                    r = 3, n-r = 0, column starts at
0 ends at 6*/
	if (n-r)>=0 then do
		q_lower = n - r;
		q_upper = 2*n;
		end;


	/*case when dimension = 4, n = 3, r = 5, n-r = -2, column starts at 0 ends at 3*3-5 = 4 */
	/*                                                    r = 6, n-r = -3, column starts at
0 ends at 3*3-6 = 3 */
	if (n-r)<0 then do
		q_lower = 0;
		q_upper = 3*n - r;
		end;

	do q = q_lower to q_upper;
		origin = (r||q);

		/*12 Legal moves: Defined on the original Hex Board*/

		/*1: Up-Left*/
		/* This means going up two squares and one step towards left diagonal */
		if r>=2 and q >=1 and r+q >= n+3 then do
			destination = (r-2||q-1);
			output;
			end;


		/*2: Up-right*/
		if r>=3 and q<= 2*n-1 and r+q >= n+2 then do
			destination = (r-3||q+1);
			output;
			end;

		/*3: right top diag and up*/
		/* This means going two steps right towards the top diagonal and then one step up */
		/*  */
		if r>=3 and q <= 2*n-2 and r+q>= n+1 then do
			destination = (r-3||q+2);
			output;
			end;

		/*4: right top diag and down*/
		if r>=2 and q <= 2*n-3 and r+q <= 3*n-1 then do
			destination = (r-2||q+3);
			output;
			end;
```

```
        /*5: right bottom diagonal up*/
        if r>=1 and q <= 2*n-3 and r+q <= 3*n-2 then do
                destination = (r-1||q+3);
                output;
                end;

        /*6: right bottom diagonal bottom*/
        if r<=2*n - 1 and q <= 2*n-2 and r+q <= 3*n-3 then do
                destination = (r+1||q+2);
                output;
                end;

        /*7: down right*/
        if r<=2*n - 2 and q <= 2*n-1 and r+q <= 3*n-3 then do
                destination = (r+2||q+1);
                output;
                end;

        /*8: down left*/
        if r<=2*n - 3 and q >= 1 and r+q <= 3*n-2 then do
                destination = (r+3||q-1);
                output;
                end;

        /*9: left bottom diagonal down*/
        if r<=2*n - 3 and q >= 2 and r+q <= 3*n-1 then do
                destination = (r+3||q-2);
                output;
                end;

        /*10: left bottom diagonal up*/
        if r<=2*n - 2 and q >= 3 and r+q >= n+1 then do
                destination = (r+2||q-3);
                output;
                end;

        /*11: left top diagonal down*/

        if r<=2*n-1 and q >= 3 and r+q >= n+2 then do
                destination = (r+1||q-3);
                output;
                end;

        /*11: left top diagonal up*/
        if r>=1 and q >= 2 and r+q >= n+3 then do
                destination = (r-1||q-2);
                output;
                end;
end;
```

```
end;

run;

proc optgraph
data_links = board;
data_links_var
from =origin
to = destination;
tsp out = closed_knights_tour4;
run;
```